
Arduino GSM Shield SIM900 

1. POPIS 

Jedná se o velice populární GSM modul, který dokáže 

fungovat jako mobilní telefon se všemi jeho 

standardními funkcemi. Spektrum využití je velmi široké, 

kromě funkcí telefonního zařízení dokáže intuitivně 

ovládat elektroniku pomocí SMS zpráv a hovorů. 

Základní charakteristika modulu: 

• Umožňuje příchozí i odchozí hovory 

• Zasílání a přijímání SMS zpráv 

• Možnost připojení k internetu (GPRS – paket pro 

přenos mobilních dat) 

• 3,5 mm jack pro mikrofon a reproduktor 

• Slot pro standardní SIM kartu 

• Odnímatelná anténa 

• funguje po celém světě (Quad-Band podporuje všechny světové frekvence) 

2. SPECIFIKACE  

Typ modulu SIM900 GPRS  třída B 

Podporované frekvence (MHz) 850 / 900 / 1800 / 1900 Pracovní teplota -40 až +85 °C 

Napájení  3,2 až 4,8 VDC Rozměry (mm) 68,3 x 53,1 

GSM standard 2/2+ Hmotnost 34 g 

 

poznámka: 

• GSM standard 2/2+ – podpora 2G a 2,5G funkcí (Mobilní data, hlasová schránka atd.) 

• GPRS třída B – zařízení může buď volat nebo pracovat s mobilními daty, nikoliv oboje zároveň. 

  



 

  

2 

ECLIPSERA s.r.o. Distributor pro ČR. 

Copyright © 2017 ECLIPSERA s.r.o. 

Verze 1.0 

3. ZAPOJENÍ 

Před samotným připojením shieldu zasuňte SIM kartu do náležitého slotu. Než tak učiníte, je třeba 

odstranit PIN, kterým je SIM karta chráněna! Tento GSM shield je kompatibilní s deskami Arduino Uno. Po 

úpravě  pořadí pinů lze použít i s deskami Arduino Mega a dalšími.  

4. UKÁZKA PROGRAMU 

Ukázka převzata z https://navody.arduino-shop.cz/navody-k-produktum/arduino-gsm-shield-

sim900.html 

// SIM 900 GSM Shield 

 

// připojení potřebných knihoven 

#include <GPRS_Shield_Arduino.h> 

#include <SoftwareSerial.h> 

// nastavení propojovacích pinů a rychlosti 

// komunikace se SIM modulem, 

// novější Shieldy používají pro 

// komunikaci TX a RX piny 7 a 8 

#define pinTX 2 

#define pinRX 3 

#define pinPower 9 

#define rychlostSim 9600 

// vytvoření proměnné s detekovaným číslem 

char mojeCislo[16] = "+420721******"; 

// vytvoření proměnných pro práci s SMS 

#define delkaZpravy 160 

char smsZprava[delkaZpravy]; 

int smsIndex = 0; 

char telCislo[16]; 

char casDatum[24]; 

// inicializace SIM900 modulu z knihovny 

GPRS sim900(pinTX, pinRX, rychlostSim); 

  

https://navody.arduino-shop.cz/navody-k-produktum/arduino-gsm-shield-sim900.html

void setup() { 

  // zahájení komunikace po sériové lince 

  Serial.begin(9600); 

  // kontrola napájení modulu SIM900, 

  // v případě vypnutého modulu ho zapneme 

  if(!sim900.checkPowerUp()) { 

    sim900.powerUpDown(pinPower); 

  } 

  // vyčkávací smyčka, dokud není správně nainicializován 

  // SIM modul 

  while(!sim900.init()) { 

      Serial.println("Chyba inicializace!"); 

      delay(1000); 

  } 

  delay(2000);   

  Serial.println("Inicializace uspesna!"); 

} 

 

void loop() { 

  // načtení a přepočet času od spuštění Arduina 

  // do tvaru 1h2m3s 

  long cas = millis()/1000; 

  long casSekundy = cas % 60; 

  long casMinuty = (cas / 60) % 60; 

  long casHodiny = cas / 3600; 

  String casKomplet = String(casHodiny); 

  casKomplet += "h"; 

  casKomplet += String(casMinuty); 

  casKomplet += "m"; 

  casKomplet += String(casSekundy); 

  casKomplet += "s"; 

  // kontrola SMS zpráv - načtení počtu nepřečtených zpráv 

  smsIndex = sim900.isSMSunread(); 

  // pokud existuje alespoň jedna nepřečtená zpráva, 

  // provedeme její přečtení a detekci textu 

  if (smsIndex > 0) { 

    // načtení SMS zprávy do proměnných ve funkci 

    sim900.readSMS(smsIndex, smsZprava, delkaZpravy, telCislo, casDatum); 

    // vymazání aktuálně přečtené zprávy 

    sim900.deleteSMS(smsIndex); 

    // vytištění informací o přijaté SMS zprávě 

    Serial.print("Od cisla: "); 

    Serial.println(telCislo);   

    Serial.print("Datum a cas: "); 

    Serial.println(casDatum); 

    Serial.print("Zprava: "); 

    Serial.println(smsZprava); 

    // kontrola přijetí SMS zprávy od našeho nastaveného čísla 

    // strcmp porovná 2 stringy a nula znamená počet odlišných znaků 

    if (strcmp(telCislo, mojeCislo)==0) { 

      // proměnná s detekovaným textem v SMS zprávě 

      char mojeZprava1[delkaZpravy] = "Ahoj"; 

      char mojeZprava2[delkaZpravy] = "Zavolej"; 

      // kontrola přijetí detekovaného textu 

      if (strcmp(smsZprava, mojeZprava1)==0) { 

        Serial.println("Prijata zprava Ahoj!"); 

      } 

      else if (strcmp(smsZprava, mojeZprava2)==0) { 

        Serial.println("Prijata zprava Zavolej, tak volam!"); 

        sim900.callUp(mojeCislo); 

      } 

      else { 

        Serial.println("Prijata neznama zprava."); 

      } 

    } 

    else { 

        Serial.println("Prijata neznama zprava od neznameho cisla."); 

    } 

  } 

  // kontrola příchozího hovoru s uložením čísla do proměnné 

  if(sim900.isCallActive(telCislo)) { 

    // vytištění informací o příchozím hovoru 

    Serial.print("Vola cislo: "); 

    Serial.println(telCislo); 

    // pauza po dobu 1 sekundy a následné zavěšení hovoru 

    delay(1000); 

    sim900.hangup(); 

    // kontrola hovoru od našeho nastaveného čísla 

    if (strcmp(telCislo, mojeCislo)==0) { 

      // vytištění informace o poslání zprávy 

      Serial.println("Zname cislo, odeslana SMS."); 

      // načtení hodnoty analogového pinu 

      // do proměnné typu String 

      String analog = String(analogRead(A0)); 

      // složení celé zprávy do proměnné odpoved 

      String odpoved = "Cas od spusteni: "; 

      odpoved += casKomplet; 

      odpoved += " vterin. A0: "; 

      odpoved += analog; 

      // převedení textu na proměnnou kompatibilní se SIM modulem 

      odpoved.toCharArray(smsZprava, 160); 

      // poslání SMS zprávy na nastavené číslo 

      sim900.sendSMS(telCislo, smsZprava); 

    } 

    else { 

      // vytištění informací o volání neznámého čísla 

      Serial.println("Nezname cislo - bez odezvy."); 

      Serial.print("Ulozene : "); 

      Serial.println(mojeCislo); 

      Serial.print("Volajici: "); 

      Serial.println(telCislo); 

    } 

  } 

  // krátká pauza před novou komunikací se SIM modulem 

  delay(10); 

} 

